

3.2 EFA/CFA on EFA/CFA

Outline


- EFA background
- EFA variations
 - ESEM, PSEM
 - Second-order, SEFA
 - Bi-factor, DSEFA
 - Target
- EFA in an SEM setting
 - MIMIC
 - **EFA/CFA on EFA/CFA**
- EFA in a multiple-group setting
 - EFA alignment
- EFA in a longitudinal setting
 - EFA longitudinal invariance testing
 - EFA longitudinal alignment
 - EFA growth modeling
- Further topics
- EFA theory

Slide 74 returns to the outline of the presentation. While MIMIC is a structural equation model, we are now turning to more general SEM settings with various combinations of EFA and CFA measurement models.

- CFA on EFA
 - CFA on EFA and X and Y
 - EFA on CFA
 - EFA on CFA and X and Y
 - EFA on EFA
-
- Factors of an EFA must be in a block by themselves, but you can have several EFA blocks

Slide 75 lists some variations of combinations of EFA and CFA measurement models in structural equation models that can be handled by Mplus. For example, there can be a CFA model for the dependent variable side of an SEM with EFA factors on the independent variable side. The reverse is also possible. Other independent or dependent variables can be added - here marked as X and Y. We will also see examples of EFA factors regressed on EFA factors, either in two different EFA blocks or within the same block. Some examples will clarify this.

CFA Regressed On EFA: UG Ex5.25



TITLE: this is an example of SEM with EFA and CFA factors with continuous factor indicators


DATA: FILE IS ex5.25.dat;

VARIABLE: NAMES = y1-y12;

MODEL: f1-f2 BY y1-y6 (*1); ! EFA factors
f3 BY y7-y9; ! CFA factor
f4 BY y10-y12; ! CFA factor
f3 ON f1-f2;
f4 ON f3;

Slide 76 shows an example of CFA factors regressed on EFA factors. This is the Mplus User's Guide example 5.25. The MODEL command shows that the f1, f2 factors are part of an EFA block, while f3 and f4 are CFA factors.

EFA Regressed On CFA: UG Ex5.25 Reversed




MODEL:

f1-f2 BY y1-y6 (*1); ! EFA factors
 f3 BY y7-y9; ! CFA factor
 f4 BY y10-y12; !CFA factor
 f1-f2 ON f3 f4;

Slide 77 shows the reverse situation with an example of EFA factors regressed on CFA factors. The EFA factors f1 and f2 are now on the dependent variable side of the model.

EFA Regressed On EFA: UG Ex5.25 Modified: Two EFA Blocks




MODEL:


f1-f2 BY y1-y6 (*1);
 f3-f4 BY y7-y12 (*2);
 f1-f2 ON f3 f4;

Slide 78 shows an example of EFA factors regressed on EFA factors. This is an example of using two EFA blocks. Here, f1 and f2 are in EFA block 1, designated by (*1), while f3 and f4 are in EFA block 2 designated by (*2).

ESEM vs PSEM



Regression between 2 blocks: Can be done by ESEM




Regression within 1 block: Can only be done by PSEM

Slide 79 shows the difference between using ESEM and using PSEM. The top part of the slide shows regression between two different EFA blocks just as in the example of the previous slide. This can be done by ESEM.

The bottom part of the slide shows that all four factors influence all 12 indicators. This represents a single EFA block with a regression between the EFA factors. Such a within-block regression can only be carried out with the help of PSEM.


The corresponding two inputs are shown on the next slide.

ESEM vs PSEM Input



Regression between 2 blocks: Can be done by ESEM

f1-f2 BY y11-y16 (*1);
 f3-f4 BY y21-y26 (*2);
 f3-f4 ON f1-f2;



Regression within 1 block: Can only be done by PSEM

f1-f4 BY y11-y26* (p1-p48); f1-f4@1;
 f3-f4 ON f1-f2;

MODEL PRIORS:

p1-p48~GEOMIN(4,1,0.0001);


Slide 80 shows the input for the two situations just discussed.

For the top case, the input shows two EFA blocks expressed in ESEM terms.

For the bottom case, a single EFA block is specified using the PSEM GEOMIN approach to EFA - which was one of the 3 EFA approaches we listed on slide 8 and used on slide 14.

The MODEL PRIORS command specifies GEOMIN rotation for the set of 4 factors. The prior variance is set at 1 and a small epsilon value is chosen. The GEOMIN prior settings are shown in the EFA Theory section of the presentation.

EFA in SEM Using PSEM: Science Example



MODEL:

```

involv BY makemeth*1 ownexp*1 choice*1 challg*0 underst*0 workhard*0 (p1-p6);
chall BY makemeth*0 ownexp*0 choice*0 challg*1 underst*1 workhard*1 (p7-p12);
involv@1 chall@1; ! 1 block: the 2 factors measured by makemeth-workhard
chall ON involv; involv ON scigra6-certsci; chall ON scigra6-certsci*0;
  
```

ANALYSIS:

```

! ITERATIONS = 10000; STARTS = 50; CONVERGENCE = 0.000001;
  
```

MODEL PRIORS:

```

p1-p12~GEOMIN(2,1,0.0001);
  
```

Slide 81 shows an example of regression within an EFA block. The example is based on the Science data used in Muthén & Asparouhov (2012). A subset of this was used as a PSEM illustration in the Asparouhov (2023) M3 workshop which you find on the next slide.

The two EFA factors challenge and involve are measured by six indicators. The challenge factor is regressed on the involve factor, that is, the model has a within-block regression like at the bottom of slide 80. Both factors are regressed on three covariates.

The input gives labels to the 12 loadings which are given GEOMIN priors. The number 2 refers to the number of factors, 1 is the prior variance, and epsilon is set to a small number which is required in this situation.

PSEM Presentations and Web Site

- Asparouhov & Muthén (2024). Penalized structural equation models. *Structural Equation Modeling*. 31, 429-454.
- Asparouhov & Muthén (2025). Methodological advances with penalized structural equation models. *Structural Equation Modeling*. 32, 688-716.
- Asparouhov (2023) M3 conference workshop: <https://www.statmodel.com/download/M3TeachingSlides.pdf>
- Asparouhov (2023) M3 conference talk: <https://www.statmodel.com/download/M3Talk.pdf>
- All of the above found at the web site **Special Mplus Topics, PSEM**: <https://www.statmodel.com/psem.shtml>

Slide 82 gives a list of references for further PSEM readings. As you see at the bottom, there is a special web site for PSEM.